

DAVID BROWN

Are We Tilting At Windmills?

The Quest For Reliable Crossovers

Moving average crossovers should be reliable and they need to have minimal lag. Here's a technique to achieve it.

by *Sylvain Vervoort*

Generally, it's a contradiction to have a smooth reliable moving average crossover on one side while having a fast reaction to price turning points on the other. "Faster" normally refers to shorter moving averages, but that in turn will result in more choppy moves, creating more false buy and sell signals. So if you want to create fast and reliable crossovers, the first thing you have to look at is a way to smooth closing

prices with as little lag as possible. In Figure 1 you can see three different averages on the daily price bars:

- The green line represents the 10-day exponential moving average (EMA). The main disadvantage seems to be that this average lags. It takes a couple of weeks for it to signal a new uptrend.
- The red line represents the 10-day TEMA (triple exponential moving average). You can see that this average responds quickly, but there's not enough smoothing. It's almost as choppy as the closing prices.

TEMA

Tema(Close, 10, Exponential)

Or a more general method:

$3 * \text{Mov}(C, 10, E) -$

$3 * \text{Mov}(\text{Mov}(C, 10, E), 10, E) + \text{Mov}(\text{Mov}(\text{Mov}(C, 10, E), 10, E), 10, E)$

where Mov = moving average, C = Closing price,

E = Exponential average

- The blue line is a 10-day TEMA based on the *heikin-ashi* closing prices. It looks good; you can tell that it is fast, and the smoothing is excellent.

TRIPLE EXPONENTIAL MOVING AVERAGE

TEMA was introduced in *Technical Analysis of STOCKS & COMMODITIES* in 1994 by Patrick Mulloy. TEMA makes use of a technique from chartist John Wilder Tukey to compensate lagging. The price data is passed through the same filter multiple times and then combined:

$$\text{TEMA} = 3 * \text{EMA} - 3 * \text{EMA}(\text{EMA}) + \text{EMA}(\text{EMA}(\text{EMA}))$$

A 10-day TEMA on closing prices in the MetaStock formula can be seen in the first sidebar, "TEMA."

HEIKIN-ASHI

Heikin-ashi, Japanese for "average bar," is a technique used in candlestick bars to better visualize price trends. It was introduced by Dan Valcu in the February 2004 issue of S&C. The heikin-ashi candlestick bars are calculated as follows:

Average price of the current bar:

$$x\text{Close} = (\text{Open} + \text{High} + \text{Low} + \text{Close}) / 4$$

Midpoint of the previous bar:

$$x\text{Open} = [x\text{Open}(\text{Previous Bar}) + \text{Close}(\text{Previous Bar})] / 2$$

Highest value in the set:

$$x\text{High} = \text{Max}(\text{High}, x\text{Open}, x\text{Close})$$

Lowest value in the set:

$$x\text{Low} = \text{Min}(\text{Low}, x\text{Open}, x\text{Close})$$




FIGURE 1: THREE DIFFERENT AVERAGES. Here you see the 10-day EMA (green), 10-day TEMA (red), and the 10-day TEMA based on the heikin-ashi closing prices.

HEIKIN-ASHI

$ha\text{Open} := (\text{Ref}((O+H+L+C)/4, -1) + \text{PREV}) / 2;$

$haC := ((O+H+L+C) /$

$4 + ha\text{Open} + \text{Max}(H, ha\text{Open}) + \text{Min}(L, ha\text{Open})) / 4;$

However, since the xClose component is always less than or equal to the high component, it is redundant within the MAX function of xHigh and a modified xHigh can be defined as (thanks for the remark made by Steve Hegji):

$$\text{modified } x\text{High} = \text{Max}(\text{High}, x\text{Open});$$

The same kind of reasoning is valid for the xLow function:

$$\text{modified } x\text{Low} = \text{Min}(\text{Low}, x\text{Open});$$

Next, I calculate the heikin-ashi closing price as the result of:

$$(x\text{Close} + x\text{Open} + x\text{High} + x\text{Low}) / 4$$

The MetaStock formula I use to calculate the heikin-ashi closing price is displayed in the second sidebar, "Heikin-ashi."

FASTER WITH ZERO LAGGING

Smoothing data with less lag and zero-lag techniques was proposed by Patrick Mulloy in the February 1994 issue of S&C and by John Ehlers in the March 2000 issue. It is a technique to compensate for the lag in moving averages.

The MetaStock code for a zero-lagging exponential moving average and TEMA can be seen in the third sidebar, "Zero-lagging exponential moving average."

In Figure 2 you can see (on the same daily chart as Figure 1) three different averages:

- The red line represents the 10-day TEMA.
- The blue one is the 10-day TEMA based on heikin-ashi closing prices.
- And the green line is a 10-day TEMA zero-lagging average based on heikin-ashi closing prices.




FIGURE 2: FASTER WITH ZERO LAGGING. The 10-day TEMA zero-lagging average (green) has faster turning points than the 10-day TEMA based on heikin-ashi closing prices (blue line).

ZERO-LAGGING EMA AND TEMA**EMA**

```

Period:= Input("What Period",1,250,10);
EMA1:= Mov(CLOSE,Period,E);
EMA2:= Mov(EMA1,Period,E);
Difference:= EMA1 - EMA2;
ZeroLagEMA:= EMA1 + Difference;
ZeroLagEMA

```

TEMA

```

Period := Input("What TEMA period? ",1,250,10);
TMA1:= Tema(CLOSE,period);
TMA2:= Tema(TMA1,period);
Difference:= TMA1 - TMA2;
ZeroLagTMA:= TMA1 + Difference;
ZeroLagTMA

```

The green line or the 10-day TEMA zero-lagging average has faster turning points than the 10-day TEMA based on heikin-ashi closing prices (blue line) without zero-lagging, and with only a fraction less smoothing.

FINDING FAST RELIABLE CROSSOVERS

It is pretty clear that the zero-lagging heikin-ashi TEMA is going to be one of the crossing averages. A zero-lagging TEMA on the median price, meaning the high price plus the low price divided by two, gives the best result in the chart used in Figure 2.

Look at Figure 3, the dashed red line is the 10-day zero-lagging heikin-ashi TEMA. The blue line is the 10-day median price zero lagging TEMA. This is the best result I could create. It has very few false crossovers.

In general, there appears to be little difference between using the median price $(high + low)/2$ or the typical price $(high + low + close)/3$. Both give similar results. For further testing I used the typical price with the MetaStock function `Typ()`. (See sidebar 4, "Moving averages used.")




FIGURE 3: FAST RELIABLE CROSSOVERS. Here you see the 10-day zero lagging heikin-ashi TEMA (red) and the 10-day median price zero lagging TEMA (blue line). There are very few false crossovers.

MOVING AVERAGES USED**Typical price zero-lagging TEMA****MetaStock formula:**

```

period := Input("Average TEMA period? ",1,100,55);
TMA1:= Tema(Typ(),period);
TMA2:= Tema(TMA1,period);
Difference:= TMA1 - TMA2;
ZeroLagTMA:= TMA1 + Difference;
ZeroLagTMA

```

Heikin-ashi zero-lagging TEMA**MetaStock formula:**

```

avg := Input("Average TEMA period? ",1,100,55);
haOpen:=(Ref((O+H+L+C)/4,-1) + PREV)/2;
haC:=((O+H+L+C)/4+haOpen+Max(H,haOpen)+Min(L,haOpen))/4;
TMA1:= Tema(haC,avg);
TMA2:= Tema(TMA1,avg);
Diff:= TMA1 - TMA2;
ZeroLagHA:= TMA1 + Diff;
ZeroLagHA

```

TESTING CROSSOVERS

In order to test the crossover, I used the following settings:

Test conditions

- The first buy signals occurred around March 2003
- End of the test period is May 16, 2007 (approximately four years)
- 211 US stocks (from a list of "10-baggers," from Thomas Maskell's series in S&C)
- \$1,000 per stock (total capital \$211,000)
- No capital, profit or loss sharing between stocks
- Long positions only
- Calculated cost: 0.1% buying; 0.1% selling

CROSSOVERS**Buying formula (MetaStock)**

```

Avg:=55; {a 55 fixed day average}
haOpen:=(Ref((O+H+L+C)/4,-1) + PREV)/2;
haC:=((O+H+L+C)/4+haOpen+Max(H,haOpen)+Min(L,haOpen))/4;
TMA1:= Tema(haC,avg);
TMA2:= Tema(TMA1,avg);
Diff:= TMA1 - TMA2;
ZIH:= TMA1 + Diff;
TMA1:= Tema(typ(),avg);
TMA2:= Tema(TMA1,avg);
Diff:= TMA1 - TMA2;
ZICI:= TMA1 + Diff;
cross(ZICI,ZIH) {crossover of the zero-lagging TEMA on the typical price and the zero-lagging TEMA of the heikin-ashi closing price}

```

Selling formula (MetaStock)

Similar to the buying formula except the last line, which makes the crossover the other way around — that is, `cross(ZIH,ZICI)`

Applying these simple techniques can boost the profit generated by the automatic system to new highs.

- Buying and selling at the closing price of the signaling day.

The MetaStock buying and selling formulas are presented in the fifth sidebar, “Crossovers.”

Result

In about four years, the trader could have ended up with a profit of \$413,537, or 196% with an average of 37 trades per stock.

ANALYZING THE RESULTS

A 55-day average (that is, a Fibonacci number) seems to give the best overall results. The same average is used on all 211 stocks, so there is no optimization.

I also conducted a test using stocks from the Dutch, French, and Belgian markets. Here also I found 55 days to be the best time setting.

Behavior in a trending market

With price moving in an uptrend or a downtrend, the buy and sell signals generated are looking quite good. In Figure 4 you can see the buy signals shown by the green up arrows and the exits shown for closing the long positions by the blue exit signs. The green line is the 55-day zero-lagging TEMA based on the typical price and the red dashed line is the 55-day zero-lagging TEMA based on the heikin-ashi closing price.

Behavior in a flat moving market

When price moves are relatively flat — that is, moving within a range — as with any crossover system you’ll see more false buy and sell signals being generated. You can see an example of this scenario in Figure 5.

The green up arrows are the buy signals and the blue exit signs show when the long positions were closed. The green line represents the 55-day zero-lagging TEMA based on the typical price and the red dashed graph, the 55-day zero-lagging TEMA based on the heikin-ashi closing price.

When price is moving in a relatively flat range, you can avoid unwanted crossings by using trendline techniques and stop-losses. You can use stop-losses or trailing stop-losses of around 10% for daily bars dependent on the volatility of the stock.

Avoiding losing trades

How do you avoid some of the losing trades you see in Figure 5? First of all, you’re going to have to trust the system. Do not enter a position when there is no signal. When there is a buy or a sell signal, analyze it thoroughly.

The next step is to accept a loss after opening a position with a stop-loss in the order of 10%. In addition, keep a trailing stop at 10%. Let’s analyze the chart in Figure 5:

- 1 = Normal exit after a very nice up move
- 2 = Normal buy after a correction — it could be the start of a continuing uptrend




FIGURE 4: BEHAVIOR IN A TRENDING MARKET. Using a 55-day periodicity, the results in an uptrend or downtrend are quite impressive.




FIGURE 5: BEHAVIOR IN A FLAT MOVING MARKET. There are a number of false buy and sell signals in a market that is moving in a relatively flat range.

- 3 = Normal exit: You were in at \$46.15 and out at \$42.69; 10% stop and a lower price than the previous low
- 4 = Normal entry; a possible end of the correction
- 5 = Do not exit. You were in at \$44.13, out at \$42.63, a loss less than 5% and still above the previous low
- 6 = If you did not exit at 5, you are still in at 6
- 7 = No exit at 7; it’s only a small loss from the entry at 4 and the low price is higher than the previous low
- 8 = No exit at 8; you were in with 4 at \$44.13. An exit here would be at \$41.07, which is still less than a 10% loss
- 9 = No entry at 9 because you are still holding your long position that you entered at 4.

THE POWER OF ANALYZING

The automatic crossover system would have presented you with four trading opportunities in this flat correction. Three of them would be losing trades.

By analyzing the signals in detail, you can end up with only one losing trade. Applying these simple techniques can certainly boost the 196% profit generated by the automatic system to new highs.

Sylvain Vervoort, living in the Flemish part of Belgium, is a retired electronics engineer studying and using technical analysis for more than 30 years. He is an independent trader, writer,

publisher, and educator in the area of technical analysis and options.

SUGGESTED READING

Ehlers, John [2000]. "Hilbert Indicators Tell You When To Trade," *Technical Analysis of STOCKS & COMMODITIES*, Volume 22: February.

Maskell, Thomas [2006]. "The Search For The 10-Bagger Begins," *Technical Analysis of STOCKS & COMMODITIES*, Volume 24: June.

Mulloy, Patrick [1994]. "Smoothing Data With Less Lag," *Technical Analysis of STOCKS & COMMODITIES*, Volume 12: February.

Valcu, Dan [2004]. "Using The Heikin-Ashi Technique," *Technical Analysis of STOCKS & COMMODITIES*, Volume 22: February.

Vervoort, Sylvain [2007]. "Trading Trendline Breaks," *Technical Analysis of STOCKS & COMMODITIES*, Volume 26: July, August, September.

S&C